BenderOrszag Ex8 p291 04-21-16

N. T. Gladd

Initialization: Be sure the files *NTGStylesheet2.nb* and *NTGUtilityFunctions.m* is are in the same directory as that from which this notebook was loaded. Then execute the cell immediately below by mousing left on the cell bar to the right of that cell and then typing “shift” + “enter”. Respond “Yes” in response to the query to evaluate initialization cells.

In[64]:=
```
SetDirectory[NotebookDirectory[]];
(* set directory where source files are located *)
SetOptions[EvaluationNotebook[], (* load the StyleSheet *)
 StyleDefinitions → Get["NTGStylesheet2.nb"]];
Get["NTGUtilityFunctions.m"]; (* Load utilities package *)
```

Original notebook *BenderOrszag Ex8 p291 09-25-15*

Purpose

Evaluation of a Bessel integral using the steepest descent method for a contour passing through a saddle point

1 Example 8

Bender and Orszag consider the example of an integral representation of a Bessel function

$$\mathcal{I}(k) = J_0(k) = \operatorname{Re}\left[\frac{1}{\pi i}\int_{-i\pi/2}^{i\pi/2} e^{i k \cosh(z)}\, dz\right]$$

Before considering the saddle point approximation, I demonstrate the integral evaluates to J_0(k),

In[66]:=
```
w1[1] = 1/(π I) Exp[ I k Cosh[z]] dz
```

Out[66]=
$$-\frac{i\, dz\, e^{i k \operatorname{Cosh}[z]}}{\pi}$$

In[67]:=
```
w1[2] = w1[1] /. {z → I y, dz → I dy} // ExpToTrig
```

Out[67]=
$$\frac{dy \operatorname{Cos}[k \operatorname{Cos}[y]]}{\pi} + \frac{i\, dy \operatorname{Sin}[k \operatorname{Cos}[y]]}{\pi}$$

In[68]:=
```
w1[3] =
 Simplify[Re[w1[2]], Assumptions → {k ∈ Reals, y ∈ Reals, dy ∈ Reals}] /. dy → 1
```

Out[68]=
$$\frac{\text{Cos}[\text{k Cos}[\text{y}]]}{\pi}$$

In[69]:=
```
w1[4] = Refine[Integrate[w1[3], {y, -π/2, π/2}], {k ∈ Reals, k > 0}]
```

Out[69]=
```
BesselJ[0, k]
```

The saddle point and associated steepest descent curves were considered in notebook *BenderOrszag Ex4-5-6-7 p289 04-18-16.* To recap

In[70]:=
```
Clear[ρ];
ρ[z_] := I Cosh[z]
```

With $\rho = \phi + \mathrm{i}\,\psi$

In[72]:=
```
w1[5] = ϕ == ComplexExpand[Re[ρ[ x + I y] ]]
```

Out[72]=
```
ϕ == -Sin[y] Sinh[x]
```

In[73]:=
```
w1[6] = ψ == ComplexExpand[Im[ρ[ x + I y] ]]
```

Out[73]=
```
ψ == Cos[y] Cosh[x]
```

In[74]:=
```
w1[7] = Solve[D[ρ[z], z] == 0, z]
```

Out[74]=
```
{{z → ConditionalExpression[2 ⅈ π C[1], C[1] ∈ Integers]},
 {z → ConditionalExpression[ⅈ π + 2 ⅈ π C[1], C[1] ∈ Integers]}}
```

In[75]:=
```
w1[8] = w1[7] /. C[1] → 0
```

Out[75]=
```
{{z → 0}, {z → ⅈ π}}
```

Because the integration path passes through it, the saddle point of interest is z_{SP} = 0. At the saddle point

In[76]:=
```
w1[9] = ρ[0]
```

Out[76]=
```
ⅈ
```

The constant ψ curves passing through the saddle point satisfy

In[77]:=
```
w1[10] = ψ == ComplexExpand[Im[ρ[ x + I y] ]] /. ψ → 1
```

Out[77]=
```
1 == Cos[y] Cosh[x]
```

Those curves are

In[78]:=
```
w1[11] = Solve[w1[10], y] /. C[1] → 0
```

Out[78]=
```
{{y → -ArcCos[Sech[x]]}, {y → ArcCos[Sech[x]]}}
```

Care with branches is required when deciding how these two expressions correspond to steepest ascent and descent curves.

In[79]:=
```
Module[{k = 2, F, yForψConstant, Curve1, Curve2, lab},
 F[z_, k_] := Exp[ I k Cosh[z]];
 yForψConstant[x_] := ArcCos[Sech[x]];
 Curve1[t_] :=
  With[{y = yForψConstant[t]}, {t, y, Abs[F[t + I y, k]]}];
 Curve2[t_] :=
  With[{y = -yForψConstant[t]}, {t, y, Abs[F[t + I y, k]]}];
 lab = Stl["Steepest assent and descent curves"];
 ParametricPlot3D[{Curve1[t], Curve2[t]},
  {t, -0.5, 0.5}, AxesLabel → {Stl["x"], Stl["y"], Stl["z"]},
  PlotLegends → {TraditionalForm[ArcCos[Sech[x]]],
    TraditionalForm[-ArcCos[Sech[x]]]}, PlotLabel → lab]]
```

Out[79]=



So, the path of steepest descent is actually defined by

$$y = \begin{pmatrix} \frac{\operatorname{sech} x}{\cos} & x \geq 0 \\ -\frac{\operatorname{sech} x}{\cos} & x < 0 \end{pmatrix}$$

In the steepest descent calculation to follow, the actual paths are approximated
I examine these paths in the vicinity of the saddle point $z_{SP} = 0$.

In[80]:=
```
w1[12] = Normal@Series[ArcCos[Sech[x]], {x, 0, 3}]
```

Out[80]=

$$\frac{\pi}{2} + \frac{1}{2}\left(-\pi + (-1)^{\text{Floor}\left[\frac{\pi - 2\,\text{Arg}[x] - \text{Arg}\left[\frac{1-\text{Sech}[x]}{x^2}\right]}{2\pi}\right]}\left(2\,x - \frac{x^3}{3}\right)\right)$$

In[81]:=
```
w1[13] = {Simplify[w1[12], {x ∈ Reals, x > 0}],
   Simplify[w1[12], {x ∈ Reals, x < 0}]} // ExpandAll
```

Out[81]=

$$\left\{x - \frac{x^3}{6},\ -x + \frac{x^3}{6}\right\}$$

Visualize these approximations

In[82]:=
```
Plot[{ArcCos[Sech[x]], -ArcCos[Sech[x]], x - x^3/6, -x + x^3/6}, {x, -1.5, 1.5},
 PlotStyle → {Black, Black, Directive[Black, Dashed], Directive[Black, Dashed]},
 AxesLabel → {Stl["x"], Stl["y"]},
 PlotLabel → Stl["ψ(x, y) = 1 curves and their approximations"]]
```

Out[82]=



The approximate steepest descent path is $y = x - \frac{x^3}{6}$

In[83]:=

```
Module[{k = 2, X = 2, Y = 2, Z = 5, δF = 0.005, image = 300, saddlePoint,
   curve1, curve2, curve1Approximate, curve2Approximate, pathIntegration,
   gCurves, gSurface, lab, F, yForψConstant, yForψConstantApproximate},
  F[z_, k_] := Exp[ I k Cosh[z]];
  yForψConstant[x_] := ArcCos[Sech[x]];
  yForψConstantApproximate[x_] := x - x^3/6;

  saddlePoint = {GREEN, PointSize[0.03], Point[{0, 0, Abs@F[0, k] + δF}]};
  curve1 =
   {GREEN, Line@Table[{x, yForψConstant[x], Abs@F[x + I yForψConstant[x] , k] + δF},
      {x, -X, X, 0.1}]};
  curve2 = {BLUE, Line@Table[{x, -yForψConstant[x],
       Abs@F[x - I yForψConstant[x] , k] + δF}, {x, -X, X, 0.1}]};
  curve1Approximate = {Directive[GREEN, Dashed],
    Line@Table[{x, yForψConstantApproximate[x],
       Abs@F[x + I yForψConstantApproximate[x] , k] + δF}, {x, -X, X, 0.1}]};
  curve2Approximate = {Directive[BLUE, Dashed],
    Line@Table[{x, -yForψConstantApproximate[x],
       Abs@F[x + I (-yForψConstantApproximate[x] ), k] + δF}, {x, -X, X, 0.1}]};
  pathIntegration = {BLACK, Line@Table[{0, y, Abs@F[I y , k] + δF},
      {y, -π/2, π/2, 0.1}]};
  lab = Stl["Saddle point and \nsteepest ascent/descent curves
      for e^(i k cosh (z))\nBlack curve is integration path"];
  gCurves = Graphics3D[{saddlePoint, curve1, curve2, pathIntegration,
     curve1Approximate, curve2Approximate},
    PlotRange → {{-X, X}, {-Y, Y}, {0, Z}}, Boxed → False, PlotLabel → lab,
    Axes → Automatic, AxesLabel → {Stl["x"], Stl["y"], Stl["F"]}];
  gSurface = Plot3D[Abs[F[x + I y, k]], {x, -2, 2}, {y, -2, 2},
    ImageSize → image, MeshFunctions -> {#3 &}, Mesh → 10,
    Boxed → False, AxesLabel → {Stl["x"], Stl["y"], Stl["|f(z)|"]} ,
    PlotRange → {{-X, X}, {-Y, Y}, {0, Z}}];
  Show[{gCurves, gSurface}]]
```

Out[83]=



For large k, the integrand is oscillatory along the integration path

In[86]:=

```
Module[{k = 10, standardArgs, F},
 F[z_, k_] := 1/(π I) Exp[I k Cosh[z]];
 Plot[{Re@F[0 + I y, k], Im@F[0 + I y, k]}, {y, -π/2, π/2},
  PlotStyle → {Black, Directive[Black, Dashed]}, AxesLabel → {Stl["y"], Stl["F"]},
  PlotLabel → Stl["e^(i k cosh (z)) along integration path"],
  Ticks → {{{-π/2, "-π/2"}, {-π/4, "-π/4"}, {0, "0"},
     {π/4, "π/4"}, {π/2, "π/2"}}, Automatic}]]
```

Out[86]=



The steepest descent approximation will be accomplished by deforming the integration path to lie along the path of steepest descent. The following plots illustrate the effect of rotation on the integrand

In[87]:=

```
With[{θ = π/2, k = 20, message = "original integration path"},
 ShowIntegrandAndPath[θ, k, message]]
```

Out[87]=



In[88]:=
```
With[{θ = π / 4, k = 20,
  message = "integration path rotated along path of steepest descent"},
 ShowIntegrandAndPath[θ, k, message]]
```

Out[88]=



In[62]:=
```
Clear[ShowIntegrandAndPath];
ShowIntegrandAndPath[θ_, k_, message_] :=
 Module[{X = 2, Y = 2, Z = 5, image = 300, r = π/2, C, lab, F, g},
  F[z_, kk_] := 1/(π I) Exp[ I kk Cosh[z]];
  lab = Stl["Integrand along contour"];
  g[1] = Plot[{Re@F[r Exp[I θ] , k], Im@F[r Exp[I θ] , k]}, {r, -π/2, π/2},
    PlotStyle → {Black, Directive[Black, Dashed]}, AxesLabel →
     {Stl["x"], Stl["y"]}, ImageSize → 300, PlotRange → All, PlotLabel → lab];
  C = {Yellow, Thickness[0.01], Arrow[{{-r Cos[θ], -r Sin[θ]},
      {r Cos[θ], r Sin[θ]}}] };
  lab = Stl@StringForm["``", message];
  g[2] = ContourPlot[Abs[F[x + I y, k]], {x, -2, 2},
    {y, -2, 2}, ImageSize → image, MeshFunctions -> {#3 &},
    Mesh → 10, AxesLabel → {Stl["x"], Stl["y"], Stl["|f(z)|"]},
    PlotRange → {{-X, X}, {-Y, Y}, {0, Z}}, Epilog → C, FrameLabel → lab];
  Grid[{{g[1], g[2]}}]]
```

2 Leading order asymptotic approximation

From above, the steepest descent path is

$$y = \begin{pmatrix} \frac{\text{sech}\, x}{\cos} & x \geq 0 \\ -\frac{\text{sech}\, x}{\cos} & x < 0 \end{pmatrix}$$

From BenderOrszag Ex4-5-6-7 p289 09-21-15, the steepest descent curve is w2[8]

```
Cos[y] Cosh[x] == 1
```

Bender and Orszag obtain a leading order asymptotic approximation by just considering y = x as the steepest descent path

In[89]:=
```
w2[1] = w1[1]
```

Out[89]=
$$-\frac{i\, dz\, e^{i\, k\, \text{Cosh}[z]}}{\pi}$$

I choose a parametric representation of z corresponding to y = x

In[90]:=
```
w2[2] = w2[1] /. {z → (1 + I) t, dz → (1 + I) dt}
```

Out[90]=
$$\frac{(1 - i)\, dt\, e^{i\, k\, \text{Cosh}[(1+i)\, t]}}{\pi}$$

For t small

In[91]:=
```
w2[3] = Normal@Series[Cosh[(1 + i) t], {t, 0, 2}]
```

Out[91]=
$$1 + i\, t^2$$

In[92]:=
```
w2[4] = w2[2] /. Cosh[(1 + i) t] → w2[3]
```

Out[92]=
$$\frac{(1 - i)\, dt\, e^{i\, k\, (1+i\, t^2)}}{\pi}$$

In[93]:=
```
w2[5] = Integrate[w2[4], {t, -∞, ∞}, Assumptions → {Re[k] > 0}] /. dt → 1
```

Out[93]=
$$\frac{(1 - i)\, e^{i\, k}}{\sqrt{k}\, \sqrt{\pi}}$$

This expression can be recast into a form that corresponds to an approximate form for J_0(k)

In[94]:=
```
w2[6] = (w2[5] // ComplexExpand) /. Arg[k] → 0 // PowerExpand
```

Out[94]=
$$\frac{\text{Cos}[k]}{\sqrt{k}\, \sqrt{\pi}} + \frac{\text{Sin}[k]}{\sqrt{k}\, \sqrt{\pi}} + i \left(-\frac{\text{Cos}[k]}{\sqrt{k}\, \sqrt{\pi}} + \frac{\text{Sin}[k]}{\sqrt{k}\, \sqrt{\pi}}\right)$$

Recall J_0(k) = Re[$\mathcal{I}$(k)]

In[95]:=
```
w2[7] = w2[6][[1 ;; 2]] // Factor
```

Out[95]=
$$\frac{\mathrm{Cos}[k] + \mathrm{Sin}[k]}{\sqrt{k}\,\sqrt{\pi}}$$

Note the identity

In[96]:=
```
w2[8] = Cos[k - π / 4] == (Cos[k - π / 4] // TrigExpand)
```

Out[96]=
$$\mathrm{Cos}\left[k - \frac{\pi}{4}\right] == \frac{\mathrm{Cos}[k]}{\sqrt{2}} + \frac{\mathrm{Sin}[k]}{\sqrt{2}}$$

In[97]:=
```
w2[9] = w2[7] /. Solve[w2[8], Cos[k]][[1]] // ExpandAll
```

Out[97]=
$$\frac{\sqrt{\frac{2}{\pi}}\,\mathrm{Cos}\left[k - \frac{\pi}{4}\right]}{\sqrt{k}}$$

So, the leading order asymptotic approximation obtained by the method of steepest descent is

$$\mathcal{I}(k) = \frac{\sqrt{\frac{2}{\pi}}\,\cos\left[k - \frac{\pi}{4}\right]}{\sqrt{k}}$$

To check, I can also calculate

In[98]:=
```
Normal@Series[BesselJ[0, k], {k, ∞, 0}]
```

Out[98]=
$$\sqrt{\frac{1}{k}}\,\sqrt{\frac{2}{\pi}}\,\mathrm{Cos}\left[k - \frac{\pi}{4}\right]$$

I visually compare the leading order approximation against Mathematica’s BesselJ0

In[99]:=

```
Module[{lab, Ɪleading},
 Ɪleading[k_] := √2 Cos[k - π / 4] / √(π k);
 lab = Stl["Asymptotic approximation compared with J₀"];
 Plot[{BesselJ[0, k], Ɪleading[k]},
  {k, 0.1, 10}, PlotStyle → {Black, Directive[Blue, Dashed]},
  AxesLabel → {Stl["k"], Stl["Ɪ(k)"]}, PlotLabel → lab, PlotLegends →
   {TraditionalForm[BesselJ[0, k]], TraditionalForm[√2 Cos[k - π / 4] / √(π k)]}]]
```

Out[99]=



3 Steepest descent in general

Bender and Orszag go on to calculate the general asymptotic expansion — a calculation requiring an insightful change of variables. Consider

$$e^{k\,\rho(z)} = e^{i\,k\cosh(z)}$$

so

$$\rho(z) = i\cosh(z)$$

The saddle point is at z = 0 and $\rho(0)$ = i. I also have

$$\rho(z) = i\cosh(z) = i\,(\cosh(x)\cos(y) + i\sinh(x)\sin(y)) =$$
$$-\sinh(x)\sin(y) + i\cosh(x)\cos(y) = \text{Re}(\rho(z)) + i\,\text{Im}(\rho(z)) = \phi(z) + i\,\psi(z)$$

The steepest descent curve is defined by Im(ρ) = ψ = constant = i, and the steepest descent equation is

$$\cosh(x)\cos(y) = 1$$

It is possible to write

$$\rho(z) = -\sinh(x)\sin(y) + i\cosh(x)\cos(y) = -\sinh(x)\sin(y) + i = i + \zeta(z)$$

BO choose to parameterize the integrand along the steepest descent curve using ζ(z) as the independent variable

$$e^{k\rho(z)} = e^{i k + k\zeta(z)} = e^{i k}\, e^{k\zeta(z)}$$

So, the change of variables is from z to

$$\zeta(z) = i\cosh(z) - i$$

I wonder who first thought of using that particular transformation

In[100]:=
```
w3[1] = ζ[z] == I Cosh[z] - I
```
Out[100]=
$$\zeta[z] == -i + i\,\mathrm{Cosh}[z]$$

In[101]:=
```
w3[2] = (D[#, z]) & /@ w3[1]
```
Out[101]=
$$\zeta'[z] == i\,\mathrm{Sinh}[z]$$

In[102]:=
```
w3[3] = Solve[Cosh[z]^2 - Sinh[z]^2 == 1, Sinh[z]][[2, 1]]
```
Out[102]=
$$\mathrm{Sinh}[z] \to \sqrt{-1 + \mathrm{Cosh}[z]^2}$$

In[103]:=
```
w3[4] = w3[3] /. Solve[w3[1], Cosh[z]][[1, 1]] // ExpandAll
```
Out[103]=
$$\mathrm{Sinh}[z] \to \sqrt{-2\,i\,\zeta[z] - \zeta[z]^2}$$

In[104]:=
```
w3[5] = dζ/dz == ζ'[z] /. (w3[2] // ER) /. w3[4]
```
Out[104]=
$$\frac{d\zeta}{dz} == i\sqrt{-2\,i\,\zeta[z] - \zeta[z]^2}$$

In[105]:=
```
w3[6] = Solve[w3[5], dz][[1, 1]] // ExpandAll
```
Out[105]=
$$dz \to -\frac{i\, d\zeta}{\sqrt{-2\,i\,\zeta[z] - \zeta[z]^2}}$$

The integrand has been transformed into

In[107]:=
```
w3[7] = Exp[I k Cosh[z]] dz /. Sol[w3[1], Cosh[z]] /. w3[6]
```

Out[107]=
$$-\frac{i\, d\zeta\, e^{i k (1-i \zeta[z])}}{\sqrt{-2 i \zeta[z]-\zeta[z]^2}}$$

Now consider ζ along the steepest descent contour. I can write the steepest descent path as

In[108]:=
```
Solve[Cosh[x] Cos[y] == 1, y][[2, 1]] /. C[1] → 0
```

Out[108]=
```
y → ArcCos[Sech[x]]
```

In[109]:=
```
Module[{F, ξ, standardArgs, g},
 F[x_] := If[x ≥ 0, ArcCos[Sech[x]], -ArcCos[Sech[x]]];
 ξ[x_] := -Sinh[x] Sin[F[x]];
 g[1] = Plot[F[x], {x, -5, 5},
   AxesLabel → {Stl["x"], Stl["y"]}, PlotLabel → Stl["steepest descent path"]];
 g[2] = Plot[ξ[x], {x, -5, 5}, AxesLabel → {Stl["x"], Stl["y"]},
   PlotLabel → Stl["ξ along steepest descent path"]];
 Grid[{{g[1]}, {g[2]}}]]
```

Out[109]=





From this figure it is clear that when z = - i $\pi/2$ is extended to -∞ - i $\pi/2$ along what is called a Sommerfeld contour we see that ζ = -∞
and, when z = i $\pi/2$ is extended to ∞ + i $\pi/2$ we see that ζ = -∞.

From these considerations

$$\mathcal{I}(k) = J_0(k) = \mathrm{Re}\Big[\frac{1}{\pi i}\int_{-i\pi/2}^{i\pi/2} \mathrm{dz}\, e^{i k \cosh(z)}\Big]$$
$$= \mathrm{Re}\Big[\frac{1}{\pi i}\int_{-i\pi/2}^{0} \mathrm{dz}\, e^{i k \cosh(z)} + \frac{1}{\pi i}\int_{0}^{i\pi/2} \mathrm{dz}\, e^{i k \cosh(z)}\Big]$$
$$= \mathrm{Re}\Big[\frac{1}{\pi i}\int_{-i\pi/2-\infty}^{0} \mathrm{dz}\, e^{i k \cosh(z)} + \frac{1}{\pi i}\int_{0}^{i\pi/2+\infty} \mathrm{dz}\, e^{i k \cosh(z)}\Big]$$
$$= \mathrm{Re}\Big[\frac{e^{i k}}{\pi i}\int_{-i\pi/2-\infty}^{0} \mathrm{dz}\, e^{k\,\zeta(z)} + \frac{e^{i k}}{\pi i}\int_{0}^{i\pi/2+\infty} \mathrm{dz}\, e^{k\,\zeta(z)}\Big]$$
$$= \mathrm{Re}\Big[\frac{e^{i k}}{\pi i}\int_{-\infty}^{0}\left(-\frac{i\,\mathrm{d}\zeta}{\sqrt{-2\,i\,\zeta[z]-\zeta[z]^2}}\right) e^{k\,\zeta(z)} + \frac{e^{i k}}{\pi i}\int_{0}^{-\infty}\left(-\frac{i\,\mathrm{d}\zeta}{\sqrt{-2\,i\,\zeta[z]-\zeta[z]^2}}\right) e^{k\,\zeta(z)}\Big]$$

Let $\zeta(z) \to -\zeta(z)$

$$\mathcal{I}(k) = \mathrm{Re}\Big[\frac{e^{i k}}{\pi i}\int_{-\infty}^{0}\left(-\frac{i\,\mathrm{d}\zeta}{\sqrt{-2\,i\,\zeta[z]-\zeta[z]^2}}\right) e^{k\,\zeta(z)} + \frac{e^{i k}}{\pi i}\int_{0}^{-\infty}\left(-\frac{i\,\mathrm{d}\zeta}{\sqrt{-2\,i\,\zeta[z]-\zeta[z]^2}}\right) e^{k\,\zeta(z)}\Big]$$
$$= \mathrm{Re}\Big[\frac{e^{i k}}{\pi i}\int_{0}^{\infty}\left(-\frac{i\,(-\mathrm{d}\zeta)}{\sqrt{2\,i\,\zeta[z]-\zeta[z]^2}}\right) e^{-k\,\zeta(z)} + \frac{e^{i k}}{\pi i}\int_{0}^{\infty}\left(-\frac{i\,(-\mathrm{d}\zeta)}{\sqrt{2\,i\,\zeta[z]-\zeta[z]^2}}\right) e^{-k\,\zeta(z)}\Big]$$
$$= \mathrm{Re}\Big[2\,\frac{e^{i k}}{\pi}\int_{0}^{\infty}\mathrm{d}\zeta\left(\frac{1}{\sqrt{2\,i\,\zeta[z]-\zeta[z]^2}}\right) e^{-k\,\zeta(z)}\Big]$$

Expanding in a power series for $\zeta \ll 1$

In[110]:=
```
w3[8] = Normal@Series[(2 Exp[I k])/π 1/Sqrt[2 I ζ - ζ^2], {ζ, 0, 3}]
```

Out[110]=
$$\frac{(1-i)\,e^{i k}}{\pi\sqrt{\zeta}} - \frac{\left(\frac{1}{4}+\frac{i}{4}\right) e^{i k}\sqrt{\zeta}}{\pi} - \frac{\left(\frac{3}{32}-\frac{3 i}{32}\right) e^{i k}\,\zeta^{3/2}}{\pi} + \frac{\left(\frac{5}{128}+\frac{5 i}{128}\right) e^{i k}\,\zeta^{5/2}}{\pi}$$

Perform term by term integration

In[111]:=
```
w3[9] = Integrate[# Exp[-k ζ], {ζ, 0, ∞}, Assumptions → {Re[k] > 0}] & /@ w3[8]
```

Out[111]=
$$\frac{\left(\frac{75}{1024}+\frac{75 i}{1024}\right) e^{i k}}{k^{7/2}\sqrt{\pi}} - \frac{\left(\frac{9}{128}-\frac{9 i}{128}\right) e^{i k}}{k^{5/2}\sqrt{\pi}} - \frac{\left(\frac{1}{8}+\frac{i}{8}\right) e^{i k}}{k^{3/2}\sqrt{\pi}} + \frac{(1-i)\,e^{i k}}{\sqrt{k}\,\sqrt{\pi}}$$

In[112]:=
```
w3[10] = (w3[9] // ComplexExpand) /. Arg[k] → 0
```

Out[112]=

$$-\frac{\text{Cos}[k]}{8\,(k^2)^{3/4}\sqrt{\pi}}+\frac{\text{Cos}[k]}{(k^2)^{1/4}\sqrt{\pi}}+\frac{75\,(k^2)^{1/4}\,\text{Cos}[k]}{1024\,k^4\sqrt{\pi}}-\frac{9\,(k^2)^{3/4}\,\text{Cos}[k]}{128\,k^4\sqrt{\pi}}+\frac{\text{Sin}[k]}{8\,(k^2)^{3/4}\sqrt{\pi}}+\frac{\text{Sin}[k]}{(k^2)^{1/4}\sqrt{\pi}}-\frac{75\,(k^2)^{1/4}\,\text{Sin}[k]}{1024\,k^4\sqrt{\pi}}-\frac{9\,(k^2)^{3/4}\,\text{Sin}[k]}{128\,k^4\sqrt{\pi}}+\mathbb{i}\left(-\frac{\text{Cos}[k]}{8\,(k^2)^{3/4}\sqrt{\pi}}-\frac{\text{Cos}[k]}{(k^2)^{1/4}\sqrt{\pi}}+\frac{75\,(k^2)^{1/4}\,\text{Cos}[k]}{1024\,k^4\sqrt{\pi}}+\frac{9\,(k^2)^{3/4}\,\text{Cos}[k]}{128\,k^4\sqrt{\pi}}-\frac{\text{Sin}[k]}{8\,(k^2)^{3/4}\sqrt{\pi}}+\frac{\text{Sin}[k]}{(k^2)^{1/4}\sqrt{\pi}}+\frac{75\,(k^2)^{1/4}\,\text{Sin}[k]}{1024\,k^4\sqrt{\pi}}-\frac{9\,(k^2)^{3/4}\,\text{Sin}[k]}{128\,k^4\sqrt{\pi}}\right)$$

I am interested in the Real part of this expression

In[113]:=
```
w3[11] = w3[10][[1 ;; -2]] // PowerExpand
```

Out[113]=

$$\frac{75\,\text{Cos}[k]}{1024\,k^{7/2}\sqrt{\pi}}-\frac{9\,\text{Cos}[k]}{128\,k^{5/2}\sqrt{\pi}}-\frac{\text{Cos}[k]}{8\,k^{3/2}\sqrt{\pi}}+\frac{\text{Cos}[k]}{\sqrt{k}\,\sqrt{\pi}}-\frac{75\,\text{Sin}[k]}{1024\,k^{7/2}\sqrt{\pi}}-\frac{9\,\text{Sin}[k]}{128\,k^{5/2}\sqrt{\pi}}+\frac{\text{Sin}[k]}{8\,k^{3/2}\sqrt{\pi}}+\frac{\text{Sin}[k]}{\sqrt{k}\,\sqrt{\pi}}$$

or

In[114]:=
```
w3[12] = Collect[w3[11], {1/√k}]
```

Out[114]=

$$\frac{\frac{75\,\text{Cos}[k]}{1024\sqrt{\pi}}-\frac{75\,\text{Sin}[k]}{1024\sqrt{\pi}}}{k^{7/2}}+\frac{-\frac{9\,\text{Cos}[k]}{128\sqrt{\pi}}-\frac{9\,\text{Sin}[k]}{128\sqrt{\pi}}}{k^{5/2}}+\frac{-\frac{\text{Cos}[k]}{8\sqrt{\pi}}+\frac{\text{Sin}[k]}{8\sqrt{\pi}}}{k^{3/2}}+\frac{\frac{\text{Cos}[k]}{\sqrt{\pi}}+\frac{\text{Sin}[k]}{\sqrt{\pi}}}{\sqrt{k}}$$

Recall the identity

In[115]:=
```
w2[8]
```

Out[115]=

$$\text{Cos}\left[k-\frac{\pi}{4}\right]==\frac{\text{Cos}[k]}{\sqrt{2}}+\frac{\text{Sin}[k]}{\sqrt{2}}$$

In[116]:=
```
w3[13] = (# √2) & /@ w2[8] // Expand // Reverse
```

Out[116]=

$$\text{Cos}[k]+\text{Sin}[k]==\sqrt{2}\,\text{Cos}\left[k-\frac{\pi}{4}\right]$$

Also note

In[117]:=
```
w3[14] = Sin[k - π / 4] == (Sin[k - π / 4] // TrigExpand)
```
Out[117]=
$$\text{Sin}\left[k-\frac{\pi}{4}\right] == -\frac{\text{Cos}[k]}{\sqrt{2}}+\frac{\text{Sin}[k]}{\sqrt{2}}$$

In[118]:=
```
w3[15] = (-# √2) & /@ w3[14] // Expand // Reverse
```
Out[118]=
$$\text{Cos}[k]-\text{Sin}[k] == -\sqrt{2}\,\text{Sin}\left[k-\frac{\pi}{4}\right]$$

Rewrite the result

In[119]:=
```
w3[16] = Factor /@ w3[12]
```
Out[119]=
$$\frac{75\,(\text{Cos}[k]-\text{Sin}[k])}{1024\,k^{7/2}\sqrt{\pi}}-\frac{\text{Cos}[k]-\text{Sin}[k]}{8\,k^{3/2}\sqrt{\pi}}-\frac{9\,(\text{Cos}[k]+\text{Sin}[k])}{128\,k^{5/2}\sqrt{\pi}}+\frac{\text{Cos}[k]+\text{Sin}[k]}{\sqrt{k}\,\sqrt{\pi}}$$

and use the trigonometric identities

In[120]:=
```
w3[17] = w3[16] /. (w3[13] // ER) /. (w3[15] // ER)
```
Out[120]=
$$\frac{\sqrt{\frac{2}{\pi}}\,\text{Cos}\left[k-\frac{\pi}{4}\right]}{\sqrt{k}}-\frac{9\,\text{Cos}\left[k-\frac{\pi}{4}\right]}{64\,k^{5/2}\sqrt{2\,\pi}}-\frac{75\,\text{Sin}\left[k-\frac{\pi}{4}\right]}{512\,k^{7/2}\sqrt{2\,\pi}}+\frac{\text{Sin}\left[k-\frac{\pi}{4}\right]}{4\,k^{3/2}\sqrt{2\,\pi}}$$

and the asymptotic expansion at this order is

In[121]:=
```
w3[18] = Collect[w3[17], {Cos[k - π/4], Sin[k - π/4]}]
```
Out[121]=
$$\left(\frac{\sqrt{\frac{2}{\pi}}}{\sqrt{k}}-\frac{9}{64\,k^{5/2}\sqrt{2\,\pi}}\right)\text{Cos}\left[k-\frac{\pi}{4}\right]+\left(-\frac{75}{512\,k^{7/2}\sqrt{2\,\pi}}+\frac{1}{4\,k^{3/2}\sqrt{2\,\pi}}\right)\text{Sin}\left[k-\frac{\pi}{4}\right]$$

Note that the leading order term $O(\frac{1}{k^{1/2}})$ agrees with the result from Section 2.